

# Oracle Triggers

- A trigger is PL/SQL code associated with a table, which performs an action when a row in a table is inserted, updated, or deleted.
- Triggers are used to implement some types of data integrity constraints that cannot be enforced at the DBMS design and implementation levels
- A trigger is a stored procedure/code block associated with a table
- Triggers specify a condition and an action to be taken whenever that condition occurs
- The DBMS automatically executes the trigger when the condition is met ("fires")
- A Trigger can be ENABLE'd or DISABLE'd via the ALTER command
  - ALTER TRIGGER *trigger\_name* ENABLE;

# Oracle Triggers - general form

CREATE [OR REPLACE] TRIGGER <trigger\_name>

{BEFORE | AFTER | INSTEAD OF }

{UPDATE | INSERT | DELETE}

[OF <attribute\_name>] ON <table\_name>

[FOR EACH ROW]

[WHEN]

DECLARE

BEGIN

.... *trigger body goes here* .....

END;

# Triggering Statement

BEFORE|AFTER INSERT|UPDATE [of colname]|DELETE ON Table

- The triggering statement specifies:
  - the type of SQL statement that fires the trigger body.
  - the possible options include DELETE, INSERT, and UPDATE. One, two, or all three of these options can be included in the triggering statement specification.
  - the table associated with the trigger.
- Column List for UPDATE
  - if a triggering statement specifies UPDATE, *an optional list of columns can be included in the triggering statement*.
  - if you include a column list, the trigger is fired on an UPDATE statement only when one of the specified columns is updated.
  - if you omit a column list, the trigger is fired when any column of the associated table is updated

# Trigger Body

**BEGIN**

.....

**END;**

- is a PL/SQL block that can include SQL and PL/SQL statements. These statements are executed if the triggering statement is issued and the trigger restriction (if included) evaluates to TRUE.
- Within a trigger body of a row trigger, the PL/SQL code and SQL statements have access to the **old** and **new** column values of the current row affected by the triggering statement.
- Two correlation names exist for every column of the table being modified: **one for the old column value** and **one for the new column value**.

# Correlation Names

- Oracle uses two correlation names in conjunction with every column value of the current row being affected by the triggering statement. These are denoted by:  
    OLD.ColumnName & NEW.ColumnName
  - For DELETE, only OLD.ColumnName is meaningful
  - For INSERT, only NEW.ColumnName is meaningful
  - For UPDATE, both are meaningful
- A colon must precede the OLD and NEW qualifiers when they are used in a trigger's body, but a colon is not allowed when using the qualifiers in the WHEN clause.
- Old and new values are available in both BEFORE and AFTER **row triggers**.

# FOR EACH ROW Option

- The FOR EACH ROW option determines whether the trigger is a row trigger or a statement trigger. If you specify FOR EACH ROW, the trigger fires once for each row of the table that is affected by the triggering statement. The absence of the FOR EACH ROW option means that the trigger fires only once for each applicable statement, but not separately for each row affected by the statement.

```
CREATE OR REPLACE TRIGGER display_salary_increase
AFTER UPDATE OF empmsal ON employee
FOR EACH ROW
WHEN (new.empmsal > 1000)
BEGIN
    DBMS_OUTPUT.PUT_LINE ('Employee: ' || :new.empno || ' Old salary: ' ||
        :old.empmsal || ' New salary: ' || :new.empmsal);
END;
```

# Statement Level Trigger

- Executed once for the whole table but will have to check all rows in the table.
- In many cases, it will be inefficient.
- **No access to the correlation values :new and :old.**


# Oracle Data FK Integrity

- Oracle offers the options:
  - UPDATE
    - no action (the default - not specified)
  - DELETE
    - no action (the default - not specified)
    - cascade
    - set null
- Subtle difference between "no action" and "restrict"
  - RESTRICT - will not allow action if child records exist, checks first
  - NO ACTION - allows action and any associated triggers, *then* checks integrity
- Databases implementations vary, for example:
  - Oracle no RESTRICT
  - IBM DB2, SQLite implement both as above





# Common use of triggers



- In the model above OWNER is the PARENT (PK end) and VEHICLE is the CHILD (FK end)
- What should the database do to maintain integrity if the user:
  - attempts to UPDATE the owner\_no of the owner (parent)
  - attempts to DELETE an owner who still has vehicles in the vehicle table
- Oracle, by default, takes the safe approach
  - UPDATE NO ACTION (no update of PK permitted if child records)
  - DELETE NO ACTION (no delete permitted if child records)
  - what if you as the developer want UPDATE CASCADE?

# Oracle Triggers

```
CREATE OR REPLACE TRIGGER Owner_Upd_Cas
BEFORE UPDATE OF owner_no ON owner
FOR EACH ROW
BEGIN
    UPDATE vehicle
    SET          owner_no = :new.owner_no
    WHERE  owner_no = :old.owner_no;
    DBMS_OUTPUT.PUT_LINE ('Corresponding owner number in the VEHICLE
table has also been updated');
END;
/
```

Implement UPDATE CASCADE rule  
OWNER 1 ---- has --- M VEHICLE  
:new.owner\_no – value of owner\_no after update  
:old.owner\_no – value of owner\_no before update

- SQL Window: To CREATE triggers, include the RUN command (/) after the last line of the file

# Common use of triggers - data integrity

- A trigger can be used to enforce user-defined integrity by triggering on a preset condition, carrying out some kind of test and then if the test fails, the trigger can raise an error (and stop the action) via a call to **`raise_application_error`**

The syntax for this call is:

```
raise_application_error(-20000, 'Error message to display');
```

the -20000 is the error number which is reported to the user, the error message is the error message the user will see. The error number can be any number less than or equal to -20000.

# Common use of triggers - data integrity - example

For example: a trigger which will ensure any unit added (ie. inserted) to the UNIT table has a unit code which starts with 'FIT'. Test your trigger and ensure it works correctly and shows your error message.

```
CREATE OR REPLACE TRIGGER check_unit_code BEFORE
    INSERT ON unit
    FOR EACH ROW
BEGIN
    IF :new.unit_code NOT LIKE 'FIT%' THEN
        raise_application_error(-20000, 'Unit code must begin with FIT');
    END IF;
END;
/
-- Test Harness
-- display before value
select * from unit;

insert into unit values ('ABC0001','Test Insert',6);

-- display after value
select * from unit;
-- closes transaction
rollback;
```

# Mutating Table

- A table that is currently being modified through an INSERT, DELETE or UPDATE statement SHOULD NOT be **read from** or **written to** because it is in a **transition state** between two stable states (before and after) where data integrity can be guaranteed.
  - Such a table is called **mutating table**.

```
CREATE OR REPLACE TRIGGER Owner_Upd_Cas BEFORE
UPDATE OF owner_no ON owner
FOR EACH ROW

DECLARE
    owner_count NUMBER;

BEGIN


    SELECT COUNT(*) INTO owner_count
    FROM owner
    WHERE owner_no = :old.owner_no;

    IF owner_count = 1 THEN
        UPDATE vehicle
        SET owner_no = :NEW.owner_no
        WHERE owner_no = :OLD.owner_no;
        DBMS_OUTPUT.PUT_LINE ('Corresponding owner number in the VEHICLE table '
        || 'has also been updated');
    END IF;


END;
```

```
update owner set owner_no = 1 where owner_no = 2
Error report -
SQL Error: ORA-04091: table LSMI1.OWNER is mutating, trigger/function may not see it
ORA-06512: at "LSMI1.OWNER_UPD_CAS", line 6
ORA-04088: error during execution of trigger 'LSMI1.OWNER_UPD_CAS'
04091. 00000 - "table %.%s is mutating, trigger/function may not see it"
*Cause:      A trigger (or a user defined plsql function that is referenced in
              this statement) attempted to look at (or modify) a table that was
              in the middle of being modified by the statement which fired it.
*Action:      Rewrite the trigger (or function) so it does not read that table.
```

# Triggers Case Study



- The student enrolment database contains two derived attributes no\_student (total number of students) and ave\_mark (average mark) .
- The total number of students is updated when **an enrolment is added or deleted**.
- The average mark is updated when **an update on attribute mark is performed**.
- For audit purpose, any deletion of enrolment needs to be recorded in an audit table. The recorded information includes the username who performed the deletion, the date and time of the deletion, the student no and unit code.



**Q5. Based on the rule to maintain the integrity of the no\_student attribute in the UNIT table as well as keeping the audit record, a trigger needs to be created for \_\_\_\_\_ table. The trigger will update a value on \_\_\_\_\_ table and insert a row to \_\_\_\_\_ table.**

- A. UNIT, ENROLMENT, AUDIT
- B. ENROLMENT, UNIT, AUDIT
- C. STUDENT, ENROLMENT, AUDIT
- D. AUDIT, UNIT, ENROLMENT



# Oracle Triggers

```
CREATE OR REPLACE TRIGGER triggername
```

```
BEFORE|AFTER  INSERT|UPDATE [of colname]|DELETE  [OR  
... ]  ON  Table
```

```
FOR EACH ROW
```

```
DECLARE
```

```
    var_name  datatype [, ...]
```

```
BEGIN
```

```
    . . . . .
```

```
END ;
```

**Q6. What would be an appropriate condition for the trigger described on the previous slide?**

- A. BEFORE INSERT OR DELETE ON enrolment.
- B. AFTER INSERT OR DELETE ON enrolment.
- C. BEFORE UPDATE OF mark ON enrolment.
- D. AFTER UPDATE OF mark ON enrolment.

```
CREATE OR REPLACE TRIGGER change_enrolment
AFTER INSERT OR DELETE ON ENROLMENT
FOR EACH ROW
DECLARE
    ???????
BEGIN
    ????????
END;
```

## Q7. What would be the logic to update the no\_student attribute in the UNIT table when a new row is inserted to ENROLMENT?

- A. UPDATE unit  
SET no\_student = no\_student + 1  
WHERE unit\_code = unit code of the inserted row
- B. UPDATE unit  
SET no\_student = (SELECT count (stu\_nbr)  
FROM enrolment  
WHERE unit\_code= unit code of the inserted row)  
WHERE unit\_code = unit code of the inserted row
- C. UPDATE unit  
SET no\_student = no\_student -1  
WHERE unit\_code = unit code of the inserted row
- D. UPDATE unit

```
CREATE OR REPLACE TRIGGER change_enrolment
AFTER INSERT OR DELETE ON ENROLMENT
FOR EACH ROW
DECLARE
    ??????
BEGIN
    IF INSERTING THEN
        UPDATE unit
        SET no_student = no_student + 1
        WHERE unit_code = :new.unit_code
    ENDIF;
    ??????
END;
```

**Q8. What would be the logic for the trigger to deal with a deletion of a row in enrolment? Assume that a table audit\_trail contains audit\_time, user, sno and unitcode attributes.**

- A. UPDATE unit  
SET no\_student = no\_student -1  
WHERE unit\_code = :old.unit\_code;
- B. INSERT INTO audit\_trail VALUES  
(SYSDATE, USER,  
:old.stu\_nbr, :old.unit\_code);
- C. UPDATE unit  
SET no\_student = no\_student – 1  
WHERE unit\_code = :new.unit\_code;
- D. a and b.
- E. b and c.

# CREATE OR REPLACE TRIGGER change\_enrolment AFTER INSERT OR DELETE ON ENROLMENT FOR EACH ROW

**BEGIN**

**IF INSERTING THEN**

**UPDATE unit**

**SET no\_student = no\_student + 1**

**WHERE unit\_code = :new.unit\_code;**

**END IF;**

**IF DELETING THEN**

**UPDATE unit**

**SET no\_student = no\_student -1**

**WHERE unit\_code = :old.unit\_code;**

**INSERT INTO audit\_trail VALUES (SYSDATE, USER,  
:old.stu\_nbr, :old.unit\_code);**

**END IF;**

**END;**

# Test Harness

- it is not sufficient to code a trigger only, a suitable test harness must be developed at the same time and used to ensure the trigger is working correctly.

```
-- display before value
select * from unit;

-- test the trigger for insertion
insert into enrolment values (11111111,'FIT2001',2013,2,null,null);

-- display after value
select * from unit;

-- test the trigger for deletion
delete from enrolment where stu_nbr = 11111111 and unit_code = 'FIT2001'and enrol_year =
2013 and enrol_semester = 2;

-- display after value
select * from unit; select * from audit_trail;
-- closes transaction
rollback;
```



## Statement Level Trigger

```
create or replace
TRIGGER DELETE_STATEMENT
AFTER DELETE ON ENROLMENT
BEGIN
    INSERT INTO enrol_history VALUES (SYSDATE, USER, 'Deleted');
END;
```

## Row Level Trigger

```
create or replace
TRIGGER DELETE_ENROLMENT
AFTER DELETE ON ENROLMENT
FOR EACH ROW
BEGIN
    INSERT INTO audit_trail VALUES
        (SYSDATE, USER, :old.stu_nbr, :old.unit_code);
END;
```

# Oracle Triggers

- Use triggers where:
  - a specific operation is performed, to ensure related actions are also performed
  - to enforce integrity where data has been denormalised
  - to maintain an audit trail
  - global operations should be performed, regardless of who performs the operation
  - they do NOT duplicate the functionality built into the DBMS
  - their size is reasonably small (< 50 - 60 lines of code)
- Do not create triggers where:
  - they are recursive
  - they modify or retrieve information from triggering tables